
Was sind Perl Source Filters?

● ermoeglichen die Anpassung oder Erweiterung 
der PerlSyntax

● erlauben komplette Umstrukturierung, Ersetzung 
oder Verschluesselung des Quellcodes

● Basis vieler Acme::* sowie Lingua::*Module im 
CPAN 

● Programme, die SourceFilter verwenden, lassen 
sich leider nicht kompilieren
– weder mit Bytecode noch CBackend

Was ist Filter::Simple?

● ein einfaches Interface zu Perl Source Filters
– stellt einige Zusatzfunktionalitaet zur Verfuegung

● erlaubt z.B. das Filtern nach “Art” des Quellcodes

– Code ohne POD und Datensektion
– Code ohne POD, Datensektion und Kommentare
– Code ohne POD, Datensektion und Quotelikes
– Code ohne POD, Datensektion, Kommentare und 

Quotelikes
– nur Quotelikes
– nur Regexe
– nur Strings

Was ist Filter::Simple?

● erlaubt das Schreiben besonders kurzer Filter
– oft nicht mehr als ein einziger Regex

● dennoch volle Flexibilitaet
● Ausfuehrung in Reihenfolge der useStatements
● viele Filter lassen sich mit no wieder deaktivieren
● mittlerweile als Teil der meisten Perl

Distributionen bereits enthalten
● entwickelt von Damian Conway

Erweiterung der Syntax

● beispielsweise durch
– neue Operatoren

– neue Statements oder Konstrukte

– Einbetten fremder Sprachelemente direkt in den 
Quellcode

● z.B. Perl6Code, XMLDaten oder SQLStatements

● haeufige Anwendung: Switch
– ergaenzt die PerlSyntax um switch()caseKonstrukte

– enthalten in den meisten PerlDistributionen



Seite 6

Perl Mongers im Ruhrgebiet

Ruhr.pm

Veit Wahlich 13. Oktober 2008

Erweiterung der Syntax

#!/usr/bin/perl

use strict;
use warnings;

use Filter::MultilineComments;

# Wir sind hier doch nicht in Bayern!
#{
Oh nein! Ein BlockKommentar!
my $a = '  Franz faehrt im komplett '
  .'verwahrlosten Taxi quer durch Bayern  ';
}#

# So ist es besser:
my $a = '  Franz faehrt im komplett '
  .'verwahrlosten Taxi quer durch den '
  .'Ruhrpott  ';

Perl kennt von Hause aus lediglich
ZeilenKommentare. Moechte man
einen ganzen Programmabschnitt
auskommentieren, vermisst man
schnell BlockKommentare, wie sie
z.B. in C ueblich sind (/*...*/).

Filter::MultilineComments
implementiert dieses Sprach
element. Da /.../ in Perl bereits
verwendet wird, greifen wir zum
Einleiten des Kommentars auf #{
und zum Schliessen auf }# 
zurueck.

Erweiterung der Syntax

use Filter::VariableModifiers;

my @data = (
    ['Original',      $a   ],
    ['Grossschrift',  $^a  ],
    ['Kleinschrift',  $_a  ],
    ['Gestutzt',      $/a  ],
    ['Kapitaelchen',  $=a  ],
    ['Rueckwaerts',   $<a  ],
    ['Kombinationen', $/^<a]
  );

Filter::VariableModifiers
implementiert Modifikatoren fuer
StringSkalare. Diese erlauben es,
beim Zugriff auf die Variable den
Inhalt in einer vom gespeicherten
Wert abweichenden Darstellung
zu erhalten.

Wird zwischen $ und Variablen
namen ein ^ gesetzt, ist der Wert
gross geschrieben, durch _ klein,
mit / wird getrimmt, = setzt den
String in Kapitaelchen und < gibt
ihn rueckwaerts zurueck. Alle
Modifikatoren sind kombinierbar.

Erweiterung der Syntax

use Filter::EmbedHTML;

# Hier regulaerer PerlCode.

%>
<! Hier beginnt HTMLCode. >
<html>
  <head>
    <title>Beispiele: Filter::Simple</title>
  </head>
  <body>
    <table>
<% foreach(@data){ %>
      <tr>
        <td><% print($_>[0]) %>: </td>
        <td><% print($_>[1]) %></td>
      </tr>
<% } %>
    </table>
  </body>
</html>
<%

# Und hier ist wieder reiner PerlCode.

Filter::EmbedHTML erlaubt es,
HTMLCode direkt in den Perl
Code einzubetten. Hier wird einfach
in ein Grundgeruest ausgegeben
sowie in einer Schleife die Ausgabe
des Arrays @data, wobei jedes
Element eine Zeile mit jeweils 2
Zellen in der Tabelle ergibt.

HTMLCode wird zwischen %> und
<% definiert, anders herum gesehen
laesst sich PerlCode zwischen <%
und %> schreiben.

Funktioniert das auch?

$ ./example.pl

<! Hier beginnt HTMLCode. >
<html>
  <head>
    <title>Beispiele mit Filter::Simple</title>
  </head>
  <body>
    <table>

      <tr>
        <td>Original: </td>
        <td>  Franz faehrt im komplett verwahrlosten Taxi quer durch den Ruhrpott  </td>
      </tr>

      <tr>
        <td>Grossschrift: </td>
        <td>  FRANZ FAEHRT IM KOMPLETT VERWAHRLOSTEN TAXI QUER DURCH DEN RUHRPOTT  </td>
      </tr>

      <tr>
        <td>Kleinschrift: </td>
        <td>  franz faehrt im komplett verwahrlosten taxi quer durch den ruhrpott  </td>
      </tr>

Funktioniert das auch?

      <tr>
        <td>Gestutzt: </td>
        <td>Franz faehrt im komplett verwahrlosten Taxi quer durch den Ruhrpott</td>
      </tr>

      <tr>
        <td>Kapitaelchen: </td>
        <td>  Franz Faehrt Im Komplett Verwahrlosten Taxi Quer Durch Den Ruhrpott  </td>
      </tr>

      <tr>
        <td>Rueckwaerts: </td>
        <td>  ttoprhuR ned hcrud reuq ixaT netsolrhawrev ttelpmok mi trheaf znarF  </td>
      </tr>

      <tr>
        <td>Kombinationen: </td>
        <td>TTOPRHUR NED HCRUD REUQ IXAT NETSOLRHAWREV TTELPMOK MI TRHEAF ZNARF</td>
      </tr>

    </table>
  </body>
</html>

Filter::MultilineComments

package Filter::MultilineComments;
use Filter::Simple;

FILTER_ONLY(code => sub{s|#\{.*?\}#||gs});

Unser FilterModul
Filter::MultilineComments.

Filtere ausschliesslich PerlCode
und ersetze dabei alle
Vorkommnisse von #{...}#
durch nichts, betrachte dabei die
gesamte Eingabe als eine einzige
Zeile.

Filter::EmbedHTML

package Filter::EmbedHTML;
use Filter::Simple;

FILTER{
    s/%>(.*?)<%
     /";print('".escapeSq($1)."');\n"/xgse;
  };

sub escapeSq($){
($_) = @_;
s/(?=[\\\'])/\\/gs;
return($_);

}

Unser FilterModul
Filter::EmbedHTML.

Filtere den gesamten Inhalt der
Datei, finde alle Vorkommnisse von
%> bis <%, speichere alles
dazwischen und ersetze diese
Vorkommnisse durch print()Befehle
mit Ausgabe der gespeicherten
Werte, diese dabei sauber escapen.

Filter::VariableModifiers

package Filter::VariableModifiers;
use Filter::Simple;

FILTER_ONLY(
    code_no_comments => \&modifiers
  );

sub modifiers{
  s/\$([<\/=\^_]+)(?=\w)
   /insertCalls($1).'$'/xge;
  return($_);
}

sub insertCalls($){
  my($mods) = @_;
  my $calls = '';
  foreach(reverse(split(//, $mods))){
    if    (/</)  { $calls .= 'Filter::VariableModifiers::reverseString ' }
    elsif (/\//) { $calls .= 'Filter::VariableModifiers::truncateString ' }
    elsif (/=/)  { $calls .= 'Filter::VariableModifiers::capitalizeString ' }
    elsif (/\^/) { $calls .= 'uc ' }
    elsif (/_/)  { $calls .= 'lc ' }
  }
  return($calls);
}

Filtere allen PerlCode (ausser
Kommentare) und ersetze dabei
alle $ gefolgt von min. einem
ModifikatorZeichen (<, /, =, ^
oder _) und einem WordZeichen
(Variablenname) durch
entsprechende Aufrufe von
ModifikatorFunktionen auf diese
Variable.



Seite 14

Perl Mongers im Ruhrgebiet

Ruhr.pm

Veit Wahlich 13. Oktober 2008

Filter::VariableModifiers

sub reverseString($){
  return(join('',
    reverse(split(//, $_[0]))));
}

sub truncateString($){
  ($_) = @_;
  s/^\s+//s;
  s/\s+$//s;
  return($_);
}

sub capitalizeString($){
  ($_) = @_;
  s/(^|\s)(\w)
   /(defined($1)?$1:'').uc($2)/xge;
  return($_);
}

reverseString() spiegelt den
uebergebenen String.

truncateString() beschneidet
den uebergebenen String.
Saemtlicher Whitespace am Anfang
und Ende wird entfernt.

capitalizeString() stellt die
Anfangsbuchstaben aller Woerter,
d.h. den ersten Buchstaben am
Anfang des Strings oder nach
Whitespace (Leerzeichen, Tab, ...),
in Grossschrift.

Quellcode-Transformation

● beispielsweise
– “Verschluesselung”/Verschleierung des Quellcodes

– Programmierung mit Sprachelementen fremder 
natuerlicher Sprachen, z.B. auf Deutsch statt Englisch

● haeufig verwendet: Lingua::Romana::Perligata
– Programmierung vollstaendig auf Latein, Syntax quasi 

ohne nichtalphabetische Zeichen
● Reihenfolge unwichtig, Bezuege durch Flexion definiert
● VariablenArten unterschieden durch Deklination
● sogar lateinische Zahlen

Mfpmmmpmfpmfppf Fppmpppmffmp!

#!/usr/bin/perl

use Filter::McCormickized;

pfmpffmffpppfmp "Mfpmmmpmfpmfppf “
  .”Pfffmfmfppff.pfmppm!\n";

no Filter::McCormickized;

print “Es geht auch normal!\n”

Ein einfaches
print “Hallo Ruhr.pm!\n”;
in KennySpeak.

Unser Modul 
Filter::McCormickized fuehrt
Programme im KennyCode aus.
Alles ab
use Filter::McCormickized;
wird als Code in KennySpeak
interpretiert, mit
no Filter::McCormickized;
schaltet man wieder zurueck zu
regulaerem PerlCode.

Filter::McCormickized

package Filter::McCormickized;
use Filter::Simple;

my $table = {
    map{
        my $i = $_;
        join('',
            map{qw(m p f)[$i/$_%3]}(9,3,1)
          ), chr(97+$i);
      }(0..25)
  };

map{
    s/(.)/uc($1)/e;
    $table>{$_} = uc($table>{lc($_)})
  } keys(%{$table});

FILTER{
    s/([MPFmpf][mpf]{2})/$table>{$1}/gse;
  };

KennySpeak ist ein ternaerer Code
bestehend aus den Elementen m, p
und f. Um die Buchstaben a bis z
abzubilden, werden Kombinationen
aus je 3 Zeichen von “mmm” fuer
“a” bis “ffp” fuer “z” verwendet.
Grossbuchstaben entsprechen
“Mmm” fuer “A” bis “Ffp” fuer “Z”.
Alle anderen Zeichen bleiben gleich.

Der Filter ersetzt saemtliche
Vorkommen von “mmm” bis “ffp”
sowie der entsprechenden Gross
schrift durch die Elemente aus der
Ersetzungstabelle.

Ein Kenny-Speak-Konverter

use strict;
use warnings;

my $i;
my $table = {
    map{
        my $i = $_;
        chr(97+$i), join('',
            map{qw(m p f)[$i/$_%3]}(9,3,1)
          );
      }(0..25)
  };
map{
    $i = $table>{$_};
    $i =~ s/(.)/uc($1)/e;
    $table>{uc($_)} = $i
  } keys(%{$table});

print("#!/usr/bin/perl\n”
  .“use Filter::McCormickized;\n");

while(<>){
  s/([az])/$table>{$1}/igse;
  print;
}

Dieses Programm funktioniert
prinzipiell genau entgegengesetzt
dem Modul
Filter::McCormickized: 

Es uebersetzt zeilenweise ein
beliebiges PerlProgramm in ein
Programm in KennySpeak.

Ein Header aus PerlShebang und
Laden des Modul wird vorangestellt.

Aufruf:
./kenny.pl <in.pl >out.pl

Wo ist das Programm?

#!/usr/bin/perl
use Filter::White;

    

  
   
 

 
 
 

  

  
 
 
 

Hier steht print “Hallo Welt”; 
 aber wo?

Der Trick:
Der gesamte Quelltext wurde in
Whitespace uebersetzt, d.h. in
Kombinationen aus “ “, \t, \n und
\f.

Unser Modul Filter::White 
wandelt den “gebleichten” Code
zurueck in PerlCode, bevor er
ausgefuehrt wird.



Seite 20

Perl Mongers im Ruhrgebiet

Ruhr.pm

Veit Wahlich 13. Oktober 2008

Filter::White

package Filter::White;
use Filter::Simple;

my $table = {
    map{
        $i = $_;
        join('',
            map{
                substr(" \t\n\f",
                  $i/4**(3$_)%4, 1);
              }(0..3)
          ), chr($i);
      }(0..255)
  };

FILTER{
    s/([ \t\f\n]{4})/$table>{$1}/gse;
  };

Unser Modul Filter::White 
uebersetzt anhand einer Ueber
setzungstabelle alle 4Tupel aus
WhitespaceZeichen [ \t\f\n] 
zurueck in je ein Byte.

Dabei ist fuer das Byte mit dem
Wert n, 0 <= n <= 255 das Zeichen
an Position m, 1 <= m <= 4
des 4Tupels q bestimmt durch
(n / 44  m) mod 4 mit (0, 1, 2, 3) :=
(“ “, \r, \n, \f), also
n = 0  >  q = (“ “, “ “, “ “, “ “),
n = 1  >  q = (“ “, “ “, “ “, \t), ...,
n = 255 > q = (\f, \f, \f, \f).

Der Konverter fuer Filter::White

use strict;
use warnings;

my $i;
my $table = [
    map{
        $i = $_;
        join('',
            map{
                substr(" \t\n\f",
                  $i/4**(3$_)%4, 1);
              }(0..3)
          );
      }(0..255)
  ];

print("#!/usr/bin/perl\n"
  ."use Filter::White;\n");

while(<>){
  s/(.)/$table>[ord($1)]/gse;
  print;
}

Dieser Konverter “bleicht” regulaere
PerlProgramme.

Er arbeitet dabei umgekehrt zu
dem FilterModul Filter::White
und stellt dem ausgegebenen Code
eine PerlShebang sowie das
Laden des FilterModuls voran.

Aufruf:
./white.pl <in.pl >out.pl

Filter::Whiter

package Filter::Whiter;
use Filter::Simple;

my $table = {
    map{
        $i = $_;
        join('',
            map{
                substr(" \t",
                  $i/2**(7$_)%2, 1);
              }(0..7)
          ), chr($i)
      }(0..255)
  };

FILTER{
    s/([ \t]{8})/$table>{$1}/gse;
  };

Leider wird der FormFeed \f von
einigen Editoren als Sonderzeichen
dargestellt, so dass der
umgewandelte Quellcode dort nicht
unsichtbar erscheint. Zudem
wuerde das Speichern mit einigen
OS den \n ein \r voranstellen und
damit das Rueckwandeln (ohne
weitere Massnahmen) behindern.

Unser Modul Filter::Whiter 
verhindert diesese Probleme, indem
es wie Acme::Bleach 8Tupel aus
“ “ und \t fuer alle Zeichen ausser
\r und \n verwendet.

Der Konverter fuer Filter::Whiter

use strict;
use warnings;

my $i;
my $table = [
    map{
        $i = $_;
        join('',
            map{
                substr(" \t",
                  $i/2**(7$_)%2, 1);
              }(0..7)
          );
      }(0..255)
  ];

print("#!/usr/bin/perl\n"
  ."use Filter::Whiter;\n");

while(<>){
  s/([^\n\r])/$table>[ord($1)]/gse;
  print;
}

Das Programm whiter.pl 
“bleicht” wie white.pl regulaere
Perl Programme, jedoch fuer das
Modul Filter::Whiter.

Ausgaben fuer Filter::Whiter
sind jedoch auch doppelt so gross
wie die fuer Filter::White, da
8Tupel statt 4Tupeln benoetigt
werden.

Aufruf:
./whiter.pl <in.pl >out.pl

Filter::WhiterShorter

package Filter::WhiterShorter;
use Filter::Simple;

my $table = {
    map{
        $i = $_;
        join('',
            map{
               substr(" \t\n",
                 $i/3**(5$_)%3, 1);
              }(0..5)
          ), chr($i);
      }(0..255)
  };

FILTER{
    s/([ \t\n]{6})/$table>{$1}/gse;
  };

Unser Modul
Filter::WhiterShorter 
bildet einen Kompromiss zwischen
Filter::White und 
Filter::Whiter.

Es verwendet 6Tupel aus “ “, \t 
und \n. So sind Ausgaben nur noch
1,5mal so gross wie mit
Filter::White, sind aber in
quasi allen Editoren unsichtbar.
Die Ausgaben sind zwar anfaellig
fuer \r\nProbleme, dies liesse
sich aber mit einem weiteren Regex
zum Entfernen aller \r beheben.

Konverter f. Filter::WhiterShorter

use strict;
use warnings;

my $i;
my $table = [
    map{
        $i = $_;
        join('',
            map{
                substr(" \t\n",
                  $i/3**(5$_)%3, 1);
              }(0..5)
          );
      }(0..255)
  ];

print("#!/usr/bin/perl\n"
  ."use Filter::WhiterShorter;\n");

while(<>){
  s/(.)/$table>[ord($1)]/gse;
  print;
}

whitershorter.pl “bleicht”
Perl Programme fuer das Modul 
Filter::WhiterShorter.

Aufruf:
./whitershorter.pl \
  <in.pl >out.pl

Vielen Dank
für Eure Aufmerksamkeit

Links

● Filter::Simple
– http://search.cpan.org/dist/Filter

Simple/lib/Filter/Simple.pm 

● Switch
– http://search.cpan.org/dist/Switch/Switch.pm

● Lingua::Romana::Perligata
– http://search.cpan.org/dist/LinguaRomana

Perligata/lib/Lingua/Romana/Perligata.pm

Links

● KennySpeak
– http://www.namesuppressed.com/kenny/

● Acme::Filter::Kenny
– http://search.cpan.org/dist/AcmeFilter

Kenny/lib/Acme/Filter/Kenny.pm

● Acme::Bleach
– http://search.cpan.org/dist/Acme

Bleach/lib/Acme/Bleach.pm



####################################################
# Filter::EmbedHTML                                #
####################################################
# Einfache Moeglichkeit, um HTML-Code einzubetten, #
# der einfach auf STDOUT ausgegeben wird. Ausgabe  #
# einleiten mit "%>", beenden mit "<%".            #
####################################################
# (C)2008 Veit Wahlich, Perl Mongers im Ruhrgebiet #
####################################################

package Filter::EmbedHTML;
use Filter::Simple;

FILTER{
		s/%>(.*?)<%/";print('".escapeSq($1)."');\n"/gse;
	};

sub escapeSq($){
	($_) = @_;
	s/(?=[\\\'])/\\/gs;
	return($_);
}



####################################################
# Filter::McCormickized                            #
####################################################
# Fuehrt Perl im Kenny-Code aus, voll kompatibel   #
# zu Acme::Filter::Kenny. Siehe auch:              #
# http://www.namesuppressed.com/kenny/             #
####################################################
# (C)2008 Veit Wahlich, Perl Mongers im Ruhrgebiet #
####################################################

package Filter::McCormickized;
use Filter::Simple;

my $table = {
		map{my $i=$_; join('', map{qw(m p f)[$i/$_%3]}(9,3,1)), chr(97+$i)}(0..25)
	};

map{s/(.)/uc($1)/e; $table->{$_} = uc($table->{lc($_)})} keys(%{$table});

FILTER{s/([MPFmpf][mpf]{2})/$table->{$1}/gse};



####################################################
# Filter::MultilineComments                        #
####################################################
# Erlaubt mehrzeilige Kommentare (Blockkommentar). #
# Kommentar einleiten "#{" und beenden mit "}#".   #
####################################################
# (C)2008 Veit Wahlich, Perl Mongers im Ruhrgebiet #
####################################################

package Filter::MultilineComments;
use Filter::Simple;

FILTER_ONLY(code => sub{s|#\{.*?\}#||gs});



####################################################
# Filter::VariableModifiers                        #
####################################################
# Implementiert Praeprozessor-Modifikatoren fuer   #
# String-Skalare. "$<a" gibt "$a" rueckwaerts aus, #
# "$/a" entfernt Whitespace vom Anfang und Ende    #
# eines Strings, "$=a" entspricht "$a" mit         #
# Kapitaelchen, "$^a" ist "$a" in Grossschrift und #
# "$_a" entspricht "$a" in Kleinschrift. Alle      #
# Modifikatoren sind frei kombinierbar.            #
####################################################
# (C)2008 Veit Wahlich, Perl Mongers im Ruhrgebiet #
####################################################

package Filter::VariableModifiers;
use Filter::Simple;

FILTER_ONLY(code_no_comments => \&modifiers);

sub modifiers{
	s/\$([<\/=\^_]+)(?=\w)/insertCalls($1).'$'/ge;
	return($_);
}

sub insertCalls($){
	my($mods) = @_;
	my $calls = '';
	foreach(reverse(split(//, $mods))){
		if    (/</)  { $calls .= 'Filter::VariableModifiers::reverseString ' }
		elsif (/\//) { $calls .= 'Filter::VariableModifiers::truncateString ' }
		elsif (/=/)  { $calls .= 'Filter::VariableModifiers::capitalizeString ' }
		elsif (/\^/) { $calls .= 'uc ' }
		elsif (/_/)  { $calls .= 'lc ' }
	}
	return($calls);
}

sub reverseString($){
	return(join('', reverse(split(//, $_[0]))));
}

sub truncateString($){
	($_) = @_;
	s/^\s+//s;
	s/\s+$//s;
	return($_);
}

sub capitalizeString($){
	($_) = @_;
	s/(^|\s)(\w)/(defined($1)?$1:'').uc($2)/ge;
	return($_);
}



####################################################
# Filter::White                                    #
####################################################
# Wandelt "gebleichten" Code in 4er-Blocks aus     #
# [ \t\n\f]-Whitespace zurueck. Regulaere Perl-    #
# Programme koennen mit ./white.pl <in.pl >out.pl  #
# umgewandelt werden.                              #
####################################################
# (C)2008 Veit Wahlich, Perl Mongers im Ruhrgebiet #
####################################################

package Filter::White;
use Filter::Simple;

my $table = {map{
		$i=$_; join('',map{substr(" \t\n\f",$i/4**(3-$_)%4,1)}(0..3)), chr($i)
	}(0..255)};

FILTER{s/([ \t\f\n]{4})/$table->{$1}/gse};



####################################################
# Filter::Whiter                                   #
####################################################
# Wie Filter::White, wandelt jedoch "gebleichten"  #
# Code aus 2er-Blocks von [ \t]-Whitespace zurueck #
# und arbeitet dabei wahrscheinlich kompatibel zu  #
# Acme::Bleach. Gebleichter Code ist dabei anders  #
# als der von Filter::White unempfindlich gegen    #
# eingestreute \r, ist jedoch doppelt so gross.    #
# Programme koennen mit ./whiter.pl <in.pl >out.pl #
# umgewandelt werden.                              #
####################################################
# (C)2008 Veit Wahlich, Perl Mongers im Ruhrgebiet #
####################################################

package Filter::Whiter;
use Filter::Simple;

my $table = {map{
		$i=$_; join('',map{substr(" \t",$i/2**(7-$_)%2,1)}(0..7)), chr($i)
	}(0..255)};

FILTER{s/([ \t]{8})/$table->{$1}/gse};



####################################################
# Filter::WhiterShorter                            #
####################################################
# Wie Filter::Whiter, wandelt jedoch "gebleichten" #
# Code aus 3er-Blocks von [ \t\n]-Whitespace       #
# zurueck. Gebleichter Code ist dabei empfindlich  #
# gegen eingestreute \r, dafuer aber deutlich      #
# kleiner als der von Filter::Whiter. Programme    #
# koennen mit ./whitershoter.pl <in.pl >out.pl     #
# umgewandelt werden.                              #
####################################################
# (C)2008 Veit Wahlich, Perl Mongers im Ruhrgebiet #
####################################################

package Filter::WhiterShorter;
use Filter::Simple;

my $table = {map{
		$i=$_; join('',map{substr(" \t\n",$i/3**(5-$_)%3,1)}(0..5)), chr($i)
	}(0..255)};

FILTER{s/([ \t\n]{6})/$table->{$1}/gse};



#!/usr/bin/perl

use strict;
use warnings;
use Filter::EmbedHTML;
use Filter::MultilineComments;
use Filter::VariableModifiers;

# Wir sind hier doch nicht in Bayern!
#{
Oh nein! Ein Block-Kommentar!
my $a = '  Franz faehrt im komplett verwahrlosten Taxi quer durch Bayern  ';
}#

# So ist es besser:
my $a = '  Franz faehrt im komplett verwahrlosten Taxi quer durch den Ruhrpott  ';

my @data = (
		['Original',      $a   ],
		['Grossschrift',  $^a  ],
		['Kleinschrift',  $_a  ],
		['Gestutzt',      $/a  ],
		['Kapitaelchen',  $=a  ],
		['Rueckwaerts',   $<a  ],
		['Kombinationen', $/^<a]
	);

%>
<!-- Hier beginnt HTML-Code. -->
<html>
	<head>
		<title>Beispiele mit Filter::Simple</title>
	</head>
	<body>
		<table>
<% foreach(@data){ %>
			<tr>
				<td><% print($_->[0]) %>: </td>
				<td><% print($_->[1]) %></td>
			</tr>
<% } %>
		</table>
	</body>
</html>
<%
# Und hier ist wieder reiner Perl-Code.



#!/usr/bin/perl
####################################################
# kenny.pl                                         #
####################################################
# Wandelt ein regulaeres Perl-Programm um in       #
# Kenny-Code-Perl.                                 #
# Die Ausgabe kann durch Filter::McCormickize      #
# weiterhin normal ausgefuehrt werden und ist auch #
# kompatibel zu Acme::Filter::Kenny.               #
# Programme koennen mit ./kenny.pl <in.pl >out.pl  #
# umgewandelt werden.                              #
####################################################
# (C)2008 Veit Wahlich, Perl Mongers im Ruhrgebiet #
####################################################

use strict;
use warnings;

my $i;
my $table = {
		map{my $i=$_; chr(97+$i), join('', map{qw(m p f)[$i/$_%3]}(9,3,1))}(0..25)
	};
map{$i=$table->{$_}; $i=~s/(.)/uc($1)/e; $table->{uc($_)}=$i} keys(%{$table});

print("#!/usr/bin/perl\nuse Filter::McCormickized;\n");
while(<>){
	s/([a-z])/$table->{$1}/igse;
	print;
}



#!/usr/bin/perl
####################################################
# white.pl                                         #
####################################################
# Wandelt ein regulaeres Perl-Programm um in       #
# "gebleichten" Code aus 4er-Blocks von [ \t\n\f]- #
# Whitespace. Die Ausgabe kann durch Filter::White #
# weiterhin normal ausgefuehrt werden.             #
# Programme koennen mit ./white.pl <in.pl >out.pl  #
# umgewandelt werden.                              #
####################################################
# (C)2008 Veit Wahlich, Perl Mongers im Ruhrgebiet #
####################################################

use strict;
use warnings;

my $i;
my $table = [
		map{$i=$_; join('',map{substr(" \t\n\f",$i/4**(3-$_)%4,1)}(0..3))}(0..255)
	];

print("#!/usr/bin/perl\nuse Filter::White;\n");
while(<>){
	s/(.)/$table->[ord($1)]/gse;
	print;
}



#!/usr/bin/perl
####################################################
# whiter.pl                                        #
####################################################
# Wandelt ein regulaeres Perl-Programm um in       #
# "gebleichten" Code aus 8er-Blocks von [ \t]-     #
# Whitespace um. Die Ausgabe kann durch            #
# Filter::Whiter weiterhin normal ausgefuehrt      #
# werden und ist (wahrscheinlich) kompatibel mit   #
# Acme::Bleach.                                    #
# Programme koennen mit ./whiter.pl <in.pl >out.pl #
# umgewandelt werden.                              #
####################################################
# (C)2008 Veit Wahlich, Perl Mongers im Ruhrgebiet #
####################################################

use strict;
use warnings;

my $i;
my $table = [
		map{$i=$_; join('',map{substr(" \t",$i/2**(7-$_)%2,1)}(0..7))}(0..255)
	];

print("#!/usr/bin/perl\nuse Filter::Whiter;\n");
while(<>){
	s/([^\n\r])/$table->[ord($1)]/gse;
	print;
}



#!/usr/bin/perl
####################################################
# whitershorter.pl                                 #
####################################################
# Wandelt ein regulaeres Perl-Programm um in       #
# "gebleichten" Code aus 6er-Blocks von [ \t\n]-   #
# Whitespace. Die Ausgabe kann durch               #
# Filter::WhiterShorter weiterhin normal           #
# ausgefuehrt werden. Programme koennen mit        #
# ./whitershorter.pl <in.pl >out.pl                #
# umgewandelt werden.                              #
####################################################
# (C)2008 Veit Wahlich, Perl Mongers im Ruhrgebiet #
####################################################

use strict;
use warnings;

my $i;
my $table = [
		map{$i=$_; join('',map{substr(" \t\n",$i/3**(5-$_)%3,1)}(0..5))}(0..255)
	];

print("#!/usr/bin/perl\nuse Filter::WhiterShorter;\n");
while(<>){
	s/(.)/$table->[ord($1)]/gse;
	print;
}


